
第13章十進制算術

計算機的使用者(人)使用十進制處
理數值。計算機必須能接受十進制
輸入; 處理完畢後, 以十進制顯示出
來。

本章介紹的指令

AAA: 加法後的ASCII調整

AAD: 除法前的ASCII調整

AAM: 乘法後的ASCII調整

AAS: 減法後的ASCII調整

DAA: 加法後的十進制調整

DAS: 減法後的十進制調整

十進資料的表示法

• 二進制加碼十進制 (Binary coded decimal, 
BCD)

• 使用四個位元來表示一位數十進數值

010: 00002, 110: 00012, 210: 00102

310: 00112, 410: 01002, 510: 01012

610: 01102, 710: 01112, 810: 10002

910: 10012

• 四個位元數值還有10102, 10112 , 11002 , 
11012 , 11102 , 11112。在BCD碼中, 它們都是
不正確的值

BCD表示法

• 未打包 (Unpacked): 一個位元組存放一位數。
放在低階四位元。高階四位元放0

• 打包 (Packed): 一個位元組存放兩位數。高階四
位元和低階四位元各放一位數

• ASCII: 30H~39H,一個位元組存放一位數。類
似未打包表示法

例子: 1527

ASCII 31 35 32 37 四個位元組

未打包 01 05 02 07 四個位元組

打包 15 27 二個位元組

AAA 和 AAS

• AAA指令: 檢查AL的右邊四位元。若在A

到F之間, 或輔助進位AF=1, 將加6到AL, 

並加1到AH, 並設CF和AF為1。AL的高階
四位元被清為0。

• AAS指令: 檢查AL的右邊四位元。若在A

到F之間, 或輔助進位AF=1, 將由AL減去6, 

並由AH減去1, 並設CF和AF為1。AL的高
階四位元被清為0。

例子: 5+2。暫存器AL放 5 的 ASCII 碼 35H, 

暫存器 BL放 2 的 ASCII 碼 32H

ADD AL,BL 35H

AAA + 32H

67H

調整後 AL 成為 07, 進位 0 加到 AH, CF=0

例子: 8+4。暫存器AL放 8 的 ASCII 碼 38H, 

暫存器 BL放 4 的 ASCII 碼 34H

ADD AL,BL 38H

AAA + 34H

6CH

調整後 AL 成為 02, 進位 1 加到 AH, CF=1

例子: 9+9。暫存器AL放 9 的 ASCII 碼 39H, 

暫存器 BL放 9 的 ASCII 碼 39H

ADD AL,BL 39H

AAA + 39H

72H

調整後 AL 成為 08, 進位 1 加到 AH, CF=1

程式例子: 十進制加法
ASCVALUE1 DB '548' ;ASCII items
ASCVALUE2 DB '724'
ASCTOTAL DB '0000'
.386 ; ...

CLC
LEA SI,ASCVALUE1+2 ;Add ASCII values:
LEA DI,ASCVALUE2+2 ;initialize addresses
LEA BX,ASCTOTAL+3 ; of ASCII numbers
MOV CX,3 ;Initialize 3 loops

A20:
MOVZX AX,[SI] ;Load ASCII byte in AX
ADC AL,[DI] ;Add (with carry)
AAA ;Adjust for ASCII
MOV [BX],AL ;Store sum
DEC SI
DEC DI
DEC BX
LOOP A20 ;Loop 3 times
MOV [BX],AH ;At end, store carry

轉成 ASCII 碼

LEA BX,ASCTOTAL+3 ;Convert ASCtotal
MOV CX,4 ;to ASCII format

A30:
OR BYTE PTR [BX],30H
DEC BX
LOOP A30 ;Loop 4 times
MOV AX,1300H ;Request display
MOV BX,0031H ;Page and attribute
LEA BP,ASCTOTAL ;ASCII line
MOV CX,4 ;Length of line
MOV DX,0824H ;Row and column
INT 10H
...

一邊計算, 一邊轉成 ASCII 碼

ADC AL,[DI] ;Add (with carry)
AAA ;Adjust for ASCII
PUSHF
OR AL,30H
POPF
MOV [BX],AL ;Store sum

OR 指令會改變旗標 (邏輯指令會使 CF,OF=0)

但 CF 在下一次iteration 中須使用到, 因此以
PUSHF和 POPF來保存

ASCII 數字的減法

例子: 9-5。ASCVAL1放 9 的 ASCII 碼 39H, 

ASCVAL2放 5 的 ASCII 碼 35H

MOV AL,ASCVAL1 39H

SUB AL,ASCVAL2 - 35H

AAS 04H

OR AL,30H

調整後 AL 仍為 04

例子: 5-9。ASCVAL1放 9 的 ASCII 碼 39H, 

ASCVAL2放 5 的 ASCII 碼 35H

MOV AL,ASCVAL2 35H

SUB AL,ASCVAL1 - 39H

AAS FCH

OR AL,30H

調整後 AL 成為 06, AH減1, CF=1

AAM 指令

將AL除以10, 商放在AH, 餘數放在AL

例子: 5*9 。暫存器AL放 5 的 ASCII 碼
35H, 暫存器 CL放 9 的 ASCII 碼 39H

AND CL,0FH

AND AL,0FH

MUL CL

AAM

OR AX,3030H

不能直接將ASCII

碼相乘

程式例子
multcand db '3783' ;ascii items
multpler db '5'
ascprod db 5 dup (0)
; ...

mov cx,4 ;initialize 4 loops
lea si,multcand+3
lea di,ascprod+4
and multpler,0fh ;Clear ASCII 3

a20:
mov al,[si] ;Load ASCII character
and al,0fh ;Clear ASCII 3
mul multpler ;Multiply
aam ;Adjust for ASCII
add al,[di] ;Add to
aaa ; stored
mov [di],al ; product
dec di
mov [di],ah
dec si
loop a20 ;Loop 4 times

; ...

lea bx,ascprod+4  ;Convert product to ASCII
mov cx,5 ;right to left, 5 bytes

a30:
or byte ptr [bx],30h
dec bx
loop a30 ;Loop 5 times
mov ax,1300h ;request display
mov bx,0031h ;Page and attribute
lea bp,ascprod;ASCII line
mov cx,5 ;Length of line
mov dx,0824h ;Row
int 10h

AAD 指令

• 在除法運算之前作。

• 將AH乘以10, 加到AL。AH 清為 0 。

例子: 28÷7 。暫存器AX放 28 的 ASCII 碼
3238H, 暫存器 CL放 7 的 ASCII 碼 37H 。

AND CL,0FH ;CL=7

AND AX,0F0FH ;AX=0208

AAD ;AX=1C

DIV CL ;AX=0004

程式例子

dividend db '3698' ;ASCII items
divisor db '4'
ascquot db 4 dup (0), '$'
; ...

mov cx,4 ;Initialize 4 loops
sub ah,ah ;clear left byte of dividend
and divisor,0fh ;Clear divisor of ASCII 3
lea si,dividend
lea di,ascquot

a20:
lodsb ;load ASCII byte
and al,0fh ;Clear ASCII 3
aad ;Adjust for divide
div divisor ;divide
stosb ;store quotient
loop a20 ;Loop 4 times

;Convert product to ASCII
lea bx,ascquot ;left to right
mov cx,4 ; 4 bytes

a30:
or byte ptr [bx],30h
inc bx
loop a30 ;Loop 4 times
mov ah,9 ;display
lea dx,ascquot ; quotient
int 21h

打包BCD資料的處理

• DAA: 若AL的低階四位元大於9, 或AF=1, 

加6到AL, 設AF=1。接著, 若AL大於99, 

或CF=1,加60H到AL, 設CF=1。

• DAS: 若AL的低階四位元大於9, 或AF=1, 

減6到AL, 設AF=1。接著, 若AL大於99, 

或CF=1,減60H到AL, 設CF=1。

程式例子

• 圖13-4, 將ASCII數值轉成Packed BCD, 而
後相加

ASCII數值轉成二進制

ASCLENTH EQU 4
ASCVALUE DB ‘3569’
BINVALUE DW 0
MULTFACT DW 1

…
MOV CX,ASCLENTH
LEA SI,ASCVALUE+3

L10:
MOV AL,[SI]
AND AX,000FH
MUL MULTFACT
ADD BINVALUE,AX
MOV AX,MULTFACT
IMUL AX,10
MOV MULTFACT,AX
DEC SI
LOOP L10

二進制數值轉成ASCII

ASCVALUE DB 4 DUP (?)
BINVALUE DW 0DF1H

…
MOV CX,10
LEA SI,ASCVALUE+3
MOV AX,BINVALUE

L20:
CMP AX,CX
JB L30
XOR DX,DX
DIV CX
OR DL,30H
MOV [SI],DL
DEC SI
JMP L20

L30:
OR AL,30H
MOV [SI],AL

四捨五入

• 運算時, 先移去小數點(小數部分位數另
外存放), 十進數值轉成二進制

• 若想將運算結果的小數部分捨去N位, 應
將結果右移N位(除以10N)

• 要作四捨五入, 應在右移前先加10N/2

例如: 1.2345 (數值是12345, 小數部分位數4)

要捨去兩位, 應先加上50(得12395)後, 再
除以100 (得123,小數部分位數剩2位)

程式例子

• 圖13-5。

• 鍵盤輸入兩個有小數點的十進數字

• 數值部分轉成二進制, 相乘

• 結果只保留2位小數點, 顯示出來

資料定義

主程式

輸入數量

轉換成二進制

相乘

轉成ASCII，前面的0變為空格

ASCII至
二進制轉換

等待按鍵、
顯示字、
清螢幕


